/*
 *  Licensed to the Apache Software Foundation (ASF) under one
 *  or more contributor license agreements.  See the NOTICE file
 *  distributed with this work for additional information
 *  regarding copyright ownership.  The ASF licenses this file
 *  to you under the Apache License, Version 2.0 (the
 *  "License"); you may not use this file except in compliance
 *  with the License.  You may obtain a copy of the License at
 *
 *    https://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing,
 *  software distributed under the License is distributed on an
 *  "AS IS" BASIS, WITHOUT WARRANTIES OR CONDITIONS OF ANY
 *  KIND, either express or implied.  See the License for the
 *  specific language governing permissions and limitations
 *  under the License.
 */

def debugArguments = [
        '-Xmx2g', '-Xdebug', '-Xnoagent', '-Djava.compiler=NONE',
        '-Xrunjdwp:transport=dt_socket,server=y,suspend=y,address=5005'
]
def java17moduleReflectionCompatibilityArguments = [
        '--add-opens=java.base/java.lang=ALL-UNNAMED',
        '--add-opens=java.base/java.util=ALL-UNNAMED'
]

tasks.withType(Test).configureEach {
    onlyIf {
        ![
                'onlyFunctionalTests',
                'onlyHibernate5Tests',
                'onlyMongodbTests',
                'skipCoreTests',
                'skipTests'
        ].find {
            project.hasProperty(it)
        }
    }

    useJUnitPlatform()
    jvmArgs += java17moduleReflectionCompatibilityArguments
    develocity {
        testRetry {
            maxRetries = 2
            maxFailures = 20
            failOnPassedAfterRetry = true
        }
    }
    testLogging {
        events('passed', 'skipped', 'failed')
        showExceptions = true
        exceptionFormat = 'full'
        showCauses = true
        showStackTraces = true
    }
    excludes = ['**/*TestCase.class', '**/*$*.class']
    maxParallelForks = configuredTestParallel
    maxHeapSize = isCiBuild ? '768m' : '1024m'
    if (System.getProperty('debug.tests')) {
        jvmArgs += debugArguments
    }
}
